
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 7.2.3.18, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
tc_7_2_3_18
Test Group:
RLC
ATS Version:
iwd-EUTRA-TVB2009-03_D10wk03
System Simulator used:
Anite Conformance Toolset
UE used:
Samsung Tanum UE.

Verification Status:
PASS
4 Corrections required for test case 7.2.3.18
4.1 Introduction

This section describes the changes required to make test case 7.2.3.18 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-TVB2009-03_D10wk03 release.

4.2 Change 1

	Testcase name
	f_TC_7_2_3_18_EUTRA

	Reason for change
	1) The t_StatusProhibit element in RLC_SS_State structure is initialised with 100ms but the default value of the t_StatusProhibit timer is 60 ms. 

	Summary of change
	1) Update the t_StatusProhibit to 60 ms after initialising the RLC variables.

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	function f_TC_7_2_3_18_EUTRA()

runs on EUTRA_PTC {

  var RLC_SS_State v_RLC_Rec;

  //EUTRA initialisation

  f_EUTRA_Init(c1);

  //RLC global initialisation

  f_InitRLC_Record(v_RLC_Rec, AM_Mode);

  //RLC test case specific initialisation

  v_RLC_Rec.UL_PDCP_SDU_size := 98;

  // Cell configuration, with one AM DRB at the SS side

  f_EUTRA_CellConfig_Def (eutra_Cell1);

  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state 3, default values with BSR and PHR disabled

  f_EUTRA_RbEst_Def_DisableBSR_PHR ( eutra_Cell1 );

  // reconfigure DRB locally to set the required test mode AM mapped to TM

  f_EUTRA_SS_DRB1_ConfigAM_TM_Mode ( eutra_Cell1);

  //closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_1DRB (v_RLC_Rec.UL_PDCP_SDU_size * 8, tsc_DRB1) );

  f_EUTRA_TestBody_Set(true);

  fl_EUTRA_7_2_3_18_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  // leaving UE test state 4

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  // postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, ATTACHED );

} // f_TC_7_2_3_18_EUTRA


After:

	function f_TC_7_2_3_18_EUTRA()

runs on EUTRA_PTC {

  var RLC_SS_State v_RLC_Rec;

  //EUTRA initialisation

  f_EUTRA_Init(c1);

  //RLC global initialisation

  f_InitRLC_Record(v_RLC_Rec, AM_Mode);

  //RLC test case specific initialisation

  v_RLC_Rec.UL_PDCP_SDU_size := 98;

  v_RLC_Rec.t_StatusProhibit := 0.06;     //NOTE 1
  // Cell configuration, with one AM DRB at the SS side

  f_EUTRA_CellConfig_Def (eutra_Cell1);

  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)

  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);

  //set null AS ciphering algorithm

  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);

  //entering UE test state 3, default values with BSR and PHR disabled

  f_EUTRA_RbEst_Def_DisableBSR_PHR ( eutra_Cell1 );

  // reconfigure DRB locally to set the required test mode AM mapped to TM

  f_EUTRA_SS_DRB1_ConfigAM_TM_Mode ( eutra_Cell1);

  //closing the test loop (mode A), entering state 4

  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_1DRB (v_RLC_Rec.UL_PDCP_SDU_size * 8, tsc_DRB1) );

  f_EUTRA_TestBody_Set(true);

  fl_EUTRA_7_2_3_18_TestBody (v_RLC_Rec);

  f_EUTRA_TestBody_Set(false);

  // leaving UE test state 4

  f_OpenUE_TestLoopMode_Deactivate_TestMode ( eutra_Cell1 );

  // postamble

  f_EUTRA_Postamble (eutra_Cell1, RRC_CONNECTED, ATTACHED );

} // f_TC_7_2_3_18_EUTRA


4.3 Change 2

	Testcase name
	fl_EUTRA_7_2_3_18_TestBody

	Reason for change
	1) After sending the status PDU at step 3, 8, 10, 15, 19, 21, 26, 28 and 29 UE will start the t_StatusProhibit timer and UE will not send any status PDU even if status PDU criteria is met until expiry of t_StatusProhibit.
2) With the default grant Size, UE will transmit RLC SDU at step 7, 8 in single RLC PDU. Similar behaviour will occur at step 13-14, 25-26, 33-34-35.
3) At step 11, SS need to transmits an AMD PDU segment of AMD PDU#3 containing SDU#3 (100 bytes) and the first 50 bytes of SDU#4 using tsc_RLC_SDU3.  However at step 9 RLC PDU is generated using variable v_AMD_SDUs3And4.
4) At step 19 UE will transmit the RLC SDU#5 with poll bit set to 1.
5) At step 21, for the NACK_SN=4 the SO field will not follow.
6) At step 21, UE will send Status PDU with 3 padding bits.

	Summary of change
	1) Before sending the new PDU add a delay equal to the t_StatusProhibit timer after receiving the status PDU.
Note: Prose CR will be raised for this in the next RAN5 meeting.

2) Modified TTCN to accept RLC SDU in a single RLC PDU.
Note: Prose CR will be raised for this in the next RAN5 meeting.
3) Used the variable v_AMD_SDUs3And4 at step 11.
4) Modified TTCN to expect RLC SDU#5 with poll bit set to 1 and transmit Status PDU.
5) Omitting the SO field for NACK_SN=4.
6) Modified TTCN to expect with 3 padding bits.

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	function fl_EUTRA_7_2_3_18_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2;

  var integer v_SDU_Size := 98;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var template (value) RLC_Status_NACK_List_Type  v_NACK_List;

  var RLC_Status_Padding_Type     v_Padding_5bits := '00000'B;

  var RLC_Status_Padding_Type     v_Padding_7bits := '0000000'B;

  var RLC_DataField_Type v_AMD_SDUs3And4;

  var RLC_DataField_Type v_AMD_SDUs8To10;

  // PDCP SDUs of size 98 will be built, appended to a PDCP header,

  // and then used as RLC SDUs

  // generate 10 RLC SDUs = PDCP PDUs

  for(i := 0; i < 10; i:= i + 1) {

    p_RLC_Rec.TxDataSize := v_SDU_Size;

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_12;

  }

  // Step 1: The SS transmits an AMD PDU containing the first half (50 bytes) of SDU#1 in its data field.

  // This PDU is in error (SN falls outside of the receiving window) and is to be discarded by the UE.

  p_RLC_Rec.AM_VTS := p_RLC_Rec.AM_VTS + tsc_AM_Window_Size + 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 50);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_NoPoll,

                  tsc_FI_StartOfSDU,

                  v_RLC_Data1 );

  // Step 2: The SS transmits an AMD PDU containing SDU#2 (100 bytes) in its data field with the P-bit set.

  p_RLC_Rec.AM_VTS := 1;

  f_TxAMD_SDU ( p_RLC_Rec, cs_TimingInfo_Now, tsc_P_Poll, tsc_RLC_SDU2 );

  // Step 3: The UE transmits a STATUS PDU with NACK_SN field indicating missing PDU#1.

  p_RLC_Rec.AM_VTA := 2;

  v_NACK_List := cr_NACK_List ( 0, tsc_E2_NoSO_Follows, omit);

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_5bits);

  // Step 4: The SS transmits an AMD PDU segment of AMD PDU#1 (AMD PDU#1 carries SDU#1) containing the

  // first 50 bytes of SDU#1 in its data field. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 5: The SS transmits an AMD PDU segment of AMD PDU#1 (AMD PDU#1 carries SDU#1) containing the last

  // 50 bytes of SDU#1 in its data field with the P-bit set. SO=50 and LSF=1. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 6: Check: Does the UE transmit a STATUS PD     with ACK_SN=2, thus acknowledging the reception of

  // PDUs with SN=0 and SN=1, and  no NACK_SN provided ?

  p_RLC_Rec.AM_VTA := 2;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

  // Step 7: Check: Does the UE transmit RLC SDU#1?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU1 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");

  // Step 8: The UE transmits RLC SDU#2.

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU2 );
  // Step 8a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 9: The SS transmits an AMD PDU segment of AMD PDU#3 (AMD PDU#3 carries SDU#3 and SDU#4) containing the last

  // 50 bytes of SDU#4 in its data field, with the P-bit set. FI=10, SO=150 and LSF=1. No header extension part is provided.

  v_AMD_SDUs3And4 := p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3] & p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4];

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (v_AMD_SDUs3And4, 150, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      150,

                      v_RLC_Data1 );

  // Step 10: The UE transmits a STATUS PDU NACK_SN field for receipt of PDU#3.

  p_RLC_Rec.AM_VTA := 3;

  v_NACK_List := cr_NACK_List ( 2,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(149, tsc_AM_SO_Size)));

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_7bits);

  // Step 11: The SS transmits an AMD PDU segment of AMD PDU#3 (AMD PDU#3 carries SDU#3 and SDU#4) containing SDU#3 (100 bytes)

  // and the first 50 bytes of SDU#4 in its data field, with the P-bit set. FI=01, SO=0 and LSF=0.

  // Header extension part present: E in fixed part header=1, E in extension part header=0, LI=100.

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 100);

  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 100, 50);

  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo_Now,

                    tsc_P_Poll,

                    tsc_FI_StartOfSDU,

                    tsc_LSF_NotLast,

                    0,

                    {v_RLC_Data1, v_RLC_Data2});

  // Step 12: Check: Does the UE transmit a STATUS PDU with ACK_SN=3?

  p_RLC_Rec.AM_VTA := 3;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

  // Step 13: The UE transmits RLC SDU#3.

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU3 );

  // Step 14: Check: Does the UE transmit RLC SDU#4?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU4 );
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 14");

  // Step 14a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 15: The SS transmits an AMD PDU segment of AMD PDU#4 (AMD PDU#4 carries SDU#5) containing the

  // first 50 bytes of SDU#5 in its data field. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 16: The SS transmits an AMD PDU segment of AMD PDU#4 (AMD PDU#4 carries SDU#5) containing the

  // first 50 bytes of SDU#5 in its data field. SO=0 and LSF=0. No header extension part is provided.

  DRB.send (cas_DRB1_RLCAmDataReq (eutra_Cell1,

                                   cs_TimingInfo_Now,

                                   p_RLC_Rec.AMD_PDU));

  // Step 17: The SS transmits an AMD PDU segment of AMD PDU#4 (AMD PDU#4 carries SDU#5) containing the last

  // 50 bytes of SDU#5 in its data field, with the P-bit set. SO=50 and LSF=1. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 18: Check: Does the UE transmit a STATUS PDU with  ACK_SN=4, thus acknowledging the reception of

  // PDUs with SN=0 to SN=3, and no NACK_SN provided ?

  p_RLC_Rec.AM_VTA := 4;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18");

  // Step 19: Check: Does the UE transmit RLC SDU#5?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU5 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");

  // Step 20: The SS transmits an AMD PDU segment of AMD PDU#6 (AMD PDU#6 carries SDU#7) containing the

  // last 50 bytes of SDU#7 in its data field, with the P-bit set. This AMD PDU segment is sent with SN 5.

  // SO=50 and LSF=1. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 21: Check: Does the UE transmit a STATUS PDU with ACK_SN=6, thus acknowledging the reception of PDUs

  // with SN=0 to SN=5, and NACK_SN=4, E1/E2 field for receipt of PDU#5 and NACK_SN=5,SOStart=0/SOEnd=49 for

  // segment 1 of PDU#6?

  p_RLC_Rec.AM_VTA := 6;

  v_NACK_List := cr_NACK_List_2NACKs ( 4,

                                       tsc_E2_SO_Follows,
                                       cr_Status_SO(int2bit(0, tsc_AM_SO_Size), tsc_SO_Special),
                                       5,

                                       tsc_E2_SO_Follows,

                                       cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(49, tsc_AM_SO_Size)) );

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_5bits);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 21");

  // Step 22: The SS transmits an AMD PDU segment of AMD PDU#6 (AMD PDU#6 carries SDU#7) containing the

  // first 50 bytes of SDU#7 in its data field. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 23: The SS transmits one AMD PDU containing SDU#6 (100 bytes) in its data field, with the P-bit set.

  p_RLC_Rec.AM_VTS := 4;

  f_TxAMD_SDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_P_Poll, tsc_RLC_SDU6);

  //Step 24: The UE transmits a STATUS PDU with ACK_SN=6, thus acknowledging the reception of PDUs with SN=0 to SN=5

  p_RLC_Rec.AM_VTA := 6;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  //Step 25: The UE transmits RLC SDU#6.

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU6 );

  //Step 26: Check: Does the UE transmit RLC SDU#7?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU7 );
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 26");

  // Step 26a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  // Step 27: The SS transmits an AMD PDU segment of AMD PDU#7 (AMD PDU#7 carries SDU#8, SDU#9 and SDU#10) containing the

  // last 20 bytes of SDU#9 and the complete SDU#10 (100 bytes) in its data field, with the P-bit set. FI=10, SO=180 and LSF=1.

  // Header extension part present: E in fixed part header=1, E in extension part header=0, LI=20.

  v_AMD_SDUs8To10 := p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8]

                     & p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9]

                     & p_RLC_Rec.RLC_DataList[tsc_RLC_SDU10];

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (v_AMD_SDUs8To10, 180, 20);

  v_RLC_Data2 := substr (v_AMD_SDUs8To10, 200, 100);

  f_TxAMD_SegList (p_RLC_Rec,

                   cs_TimingInfo_Now,

                   tsc_P_Poll,

                   tsc_FI_EndOfSDU,

                   tsc_LSF_Last,

                   180,

                   {v_RLC_Data1, v_RLC_Data2});

  // Step 28: The UE transmits a STATUS PDU NACK_SN field for receipt of PDU#7.

  p_RLC_Rec.AM_VTA := 7;

  v_NACK_List := cr_NACK_List ( 6,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(179, tsc_AM_SO_Size)));

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_7bits);

  // Step 29: The SS transmits an AMD PDU segment of AMD PDU#7 (AMD PDU#7 carries SDU#8, SDU#9 and SDU#10) containing the

  // last 20 bytes of SDU#8 and the complete SDU#9 in its data field, with the P-bit set. FI=10, SO=80 and LSF=0.

  // Header extension part present: E in fixed part header=1, E in extension part header=0, LI=20.

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (v_AMD_SDUs8To10, 80, 20);

  v_RLC_Data2 := substr (v_AMD_SDUs8To10, 100, 100);

  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo_Now,

                    tsc_P_Poll,

                    tsc_FI_EndOfSDU,

                    tsc_LSF_NotLast,

                    80,

                    {v_RLC_Data1, v_RLC_Data2});

  // Step 30: The UE transmits a STATUS PDU NACK_SN field for receipt of PDU#7.

  p_RLC_Rec.AM_VTA := 7;

  v_NACK_List := cr_NACK_List ( 6,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(79, tsc_AM_SO_Size)));

  f_RxSTATUS_PDU_NACK (p_RLC_Rec, v_NACK_List, v_Padding_7bits);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 30");

  // Step 31: The SS transmits an AMD PDU segment of AMD PDU#7 (AMD PDU#7 carries SDU#8, SDU#9 and SDU#10) containing the

  // first 80 bytes of SDU#8 in its data field, with the P-bit set. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (v_AMD_SDUs8To10, 0, 80);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1 );

  // Step 32: Check: Does the UE transmit a STATUS PDU with  ACK_SN=7, thus acknowledging the reception of PDUs with SN=0 to SN=6,

  // and no NACK_SN provided ?

  p_RLC_Rec.AM_VTA := 7;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 32");

  // Step 33: Check: Does the UE transmit RLC SDU#8?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU8 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 33");

  // Step 34: Check: Does the UE transmit RLC SDU#9?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_NoPoll, tsc_RLC_SDU9 );

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 34");

  // Step 35: Check: Does the UE transmit RLC SDU#10?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU10 );
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 35");

  // Step 35a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

} // fl_EUTRA_7_2_3_18_TestBody


After:

	function fl_EUTRA_7_2_3_18_TestBody(inout RLC_SS_State p_RLC_Rec)

runs on EUTRA_PTC {

  var integer i;

  var RLC_DataField_Type v_RLC_Data1, v_RLC_Data2, v_RLC_Data3;        // NOTE 2
  var integer v_SDU_Size := 98;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var template (value) RLC_Status_NACK_List_Type  v_NACK_List;

  var RLC_Status_Padding_Type     v_Padding_3bits := '000'B;               // NOTE 6
  var RLC_Status_Padding_Type     v_Padding_5bits := '00000'B;

  var RLC_Status_Padding_Type     v_Padding_7bits := '0000000'B;

  var RLC_DataField_Type v_AMD_SDUs3And4;

  var RLC_DataField_Type v_AMD_SDUs8To10;

  // PDCP SDUs of size 98 will be built, appended to a PDCP header,

  // and then used as RLC SDUs

  // generate 10 RLC SDUs = PDCP PDUs

  for(i := 0; i < 10; i:= i + 1) {

    p_RLC_Rec.TxDataSize := v_SDU_Size;

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12 ), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod tsc_Maximum_PDCP_SN_12;

  }

  // Step 1: The SS transmits an AMD PDU containing the first half (50 bytes) of SDU#1 in its data field.

  // This PDU is in error (SN falls outside of the receiving window) and is to be discarded by the UE.

  p_RLC_Rec.AM_VTS := p_RLC_Rec.AM_VTS + tsc_AM_Window_Size + 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 50);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_NoPoll,

                  tsc_FI_StartOfSDU,

                  v_RLC_Data1 );

  // Step 2: The SS transmits an AMD PDU containing SDU#2 (100 bytes) in its data field with the P-bit set.

  p_RLC_Rec.AM_VTS := 1;

  f_TxAMD_SDU ( p_RLC_Rec, cs_TimingInfo_Now, tsc_P_Poll, tsc_RLC_SDU2 );

  // Step 3: The UE transmits a STATUS PDU with NACK_SN field indicating missing PDU#1.

  p_RLC_Rec.AM_VTA := 2;

  v_NACK_List := cr_NACK_List ( 0, tsc_E2_NoSO_Follows, omit);

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_5bits);

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1 
  // Step 4: The SS transmits an AMD PDU segment of AMD PDU#1 (AMD PDU#1 carries SDU#1) containing the

  // first 50 bytes of SDU#1 in its data field. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 5: The SS transmits an AMD PDU segment of AMD PDU#1 (AMD PDU#1 carries SDU#1) containing the last

  // 50 bytes of SDU#1 in its data field with the P-bit set. SO=50 and LSF=1. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 6: Check: Does the UE transmit a STATUS PD     with ACK_SN=2, thus acknowledging the reception of

  // PDUs with SN=0 and SN=1, and  no NACK_SN provided ?

  p_RLC_Rec.AM_VTA := 2;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 6");

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 100);                   // NOTE 2
  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 0, 100);                   // NOTE 2
  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (100),

               {v_RLC_Data1, v_RLC_Data2});                                    // NOTE 2
  // Step 8a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 9: The SS transmits an AMD PDU segment of AMD PDU#3 (AMD PDU#3 carries SDU#3 and SDU#4) containing the last

  // 50 bytes of SDU#4 in its data field, with the P-bit set. FI=10, SO=150 and LSF=1. No header extension part is provided.

  v_AMD_SDUs3And4 := p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3] & p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4];

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (v_AMD_SDUs3And4, 150, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      150,

                      v_RLC_Data1 );

  // Step 10: The UE transmits a STATUS PDU NACK_SN field for receipt of PDU#3.

  p_RLC_Rec.AM_VTA := 3;

  v_NACK_List := cr_NACK_List ( 2,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(149, tsc_AM_SO_Size)));

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_7bits);

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 11: The SS transmits an AMD PDU segment of AMD PDU#3 (AMD PDU#3 carries SDU#3 and SDU#4) containing SDU#3 (100 bytes)

  // and the first 50 bytes of SDU#4 in its data field, with the P-bit set. FI=01, SO=0 and LSF=0.

  // Header extension part present: E in fixed part header=1, E in extension part header=0, LI=100.

  p_RLC_Rec.AM_VTS := 2;

  v_RLC_Data1 := substr (v_AMD_SDUs3And4, 0, 100);                                    // NOTE 3
  v_RLC_Data2 := substr (v_AMD_SDUs3And4, 100, 50);                                  // NOTE 3 
  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo_Now,

                    tsc_P_Poll,

                    tsc_FI_StartOfSDU,

                    tsc_LSF_NotLast,

                    0,

                    {v_RLC_Data1, v_RLC_Data2});
  // Step 12: Check: Does the UE transmit a STATUS PDU with ACK_SN=3?

  p_RLC_Rec.AM_VTA := 3;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 100);         // NOTE 2
  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU4], 0, 100);         // NOTE 2
  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (100),

               {v_RLC_Data1, v_RLC_Data2});                                    // NOTE 2
  // Step 14a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 15: The SS transmits an AMD PDU segment of AMD PDU#4 (AMD PDU#4 carries SDU#5) containing the

  // first 50 bytes of SDU#5 in its data field. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 16: The SS transmits an AMD PDU segment of AMD PDU#4 (AMD PDU#4 carries SDU#5) containing the

  // first 50 bytes of SDU#5 in its data field. SO=0 and LSF=0. No header extension part is provided.

  DRB.send (cas_DRB1_RLCAmDataReq (eutra_Cell1,

                                   cs_TimingInfo_Now,

                                   p_RLC_Rec.AMD_PDU));

  // Step 17: The SS transmits an AMD PDU segment of AMD PDU#4 (AMD PDU#4 carries SDU#5) containing the last

  // 50 bytes of SDU#5 in its data field, with the P-bit set. SO=50 and LSF=1. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 3;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU5], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 18: Check: Does the UE transmit a STATUS PDU with  ACK_SN=4, thus acknowledging the reception of

  // PDUs with SN=0 to SN=3, and no NACK_SN provided ?

  p_RLC_Rec.AM_VTA := 4;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18");

  // Step 19: Check: Does the UE transmit RLC SDU#5?

  f_RxAMD_SDU ( p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU5 );                                    // NOTE 4
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");

  // Step 19a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);                                    // NOTE 4
  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 20: The SS transmits an AMD PDU segment of AMD PDU#6 (AMD PDU#6 carries SDU#7) containing the

  // last 50 bytes of SDU#7 in its data field, with the P-bit set. This AMD PDU segment is sent with SN 5.

  // SO=50 and LSF=1. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 50, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_EndOfSDU,

                      tsc_LSF_Last,

                      50,

                      v_RLC_Data1);

  // Step 21: Check: Does the UE transmit a STATUS PDU with ACK_SN=6, thus acknowledging the reception of PDUs

  // with SN=0 to SN=5, and NACK_SN=4, E1/E2 field for receipt of PDU#5 and NACK_SN=5,SOStart=0/SOEnd=49 for

  // segment 1 of PDU#6?

  p_RLC_Rec.AM_VTA := 6;

  v_NACK_List := cr_NACK_List_2NACKs ( 4,

                                       tsc_E2_NoSO_Follows,                                    // NOTE 5
                                       omit,                                                                  // NOTE 5
                                       5,

                                       tsc_E2_SO_Follows,

                                       cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(49, tsc_AM_SO_Size)) );

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_3bits );               // NOTE 6
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 21");

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 22: The SS transmits an AMD PDU segment of AMD PDU#6 (AMD PDU#6 carries SDU#7) containing the

  // first 50 bytes of SDU#7 in its data field. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 5;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 50);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_NoPoll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1);

  // Step 23: The SS transmits one AMD PDU containing SDU#6 (100 bytes) in its data field, with the P-bit set.

  p_RLC_Rec.AM_VTS := 4;

  f_TxAMD_SDU (p_RLC_Rec, cs_TimingInfo_Now, tsc_P_Poll, tsc_RLC_SDU6);

  //Step 24: The UE transmits a STATUS PDU with ACK_SN=6, thus acknowledging the reception of PDUs with SN=0 to SN=5

  p_RLC_Rec.AM_VTA := 6;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU6], 0, 100);                 // NOTE 2
  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU7], 0, 100);                 // NOTE 2
  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_1LI (100),

               {v_RLC_Data1, v_RLC_Data2});                                    // NOTE 2
  // Step 26a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 27: The SS transmits an AMD PDU segment of AMD PDU#7 (AMD PDU#7 carries SDU#8, SDU#9 and SDU#10) containing the

  // last 20 bytes of SDU#9 and the complete SDU#10 (100 bytes) in its data field, with the P-bit set. FI=10, SO=180 and LSF=1.

  // Header extension part present: E in fixed part header=1, E in extension part header=0, LI=20.

  v_AMD_SDUs8To10 := p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8]

                     & p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9]

                     & p_RLC_Rec.RLC_DataList[tsc_RLC_SDU10];

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (v_AMD_SDUs8To10, 180, 20);

  v_RLC_Data2 := substr (v_AMD_SDUs8To10, 200, 100);

  f_TxAMD_SegList (p_RLC_Rec,

                   cs_TimingInfo_Now,

                   tsc_P_Poll,

                   tsc_FI_EndOfSDU,

                   tsc_LSF_Last,

                   180,

                   {v_RLC_Data1, v_RLC_Data2});

  // Step 28: The UE transmits a STATUS PDU NACK_SN field for receipt of PDU#7.

  p_RLC_Rec.AM_VTA := 7;

  v_NACK_List := cr_NACK_List ( 6,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(179, tsc_AM_SO_Size)));

  f_RxSTATUS_PDU_NACK(p_RLC_Rec, v_NACK_List, v_Padding_7bits);

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 29: The SS transmits an AMD PDU segment of AMD PDU#7 (AMD PDU#7 carries SDU#8, SDU#9 and SDU#10) containing the

  // last 20 bytes of SDU#8 and the complete SDU#9 in its data field, with the P-bit set. FI=10, SO=80 and LSF=0.

  // Header extension part present: E in fixed part header=1, E in extension part header=0, LI=20.

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (v_AMD_SDUs8To10, 80, 20);

  v_RLC_Data2 := substr (v_AMD_SDUs8To10, 100, 100);

  f_TxAMD_SegList ( p_RLC_Rec,

                    cs_TimingInfo_Now,

                    tsc_P_Poll,

                    tsc_FI_EndOfSDU,

                    tsc_LSF_NotLast,

                    80,

                    {v_RLC_Data1, v_RLC_Data2});

  // Step 30: The UE transmits a STATUS PDU NACK_SN field for receipt of PDU#7.

  p_RLC_Rec.AM_VTA := 7;

  v_NACK_List := cr_NACK_List ( 6,

                                tsc_E2_SO_Follows,

                                cr_Status_SO(int2bit(0, tsc_AM_SO_Size), int2bit(79, tsc_AM_SO_Size)));

  f_RxSTATUS_PDU_NACK (p_RLC_Rec, v_NACK_List, v_Padding_7bits);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 30");

  f_Delay (p_RLC_Rec.t_StatusProhibit);                                    // NOTE1
  // Step 31: The SS transmits an AMD PDU segment of AMD PDU#7 (AMD PDU#7 carries SDU#8, SDU#9 and SDU#10) containing the

  // first 80 bytes of SDU#8 in its data field, with the P-bit set. SO=0 and LSF=0. No header extension part is provided.

  p_RLC_Rec.AM_VTS := 6;

  v_RLC_Data1 := substr (v_AMD_SDUs8To10, 0, 80);

  f_TxAMD_PDU_OneSeg (p_RLC_Rec,

                      cs_TimingInfo_Now,

                      tsc_P_Poll,

                      tsc_FI_StartOfSDU,

                      tsc_LSF_NotLast,

                      0,

                      v_RLC_Data1 );

  // Step 32: Check: Does the UE transmit a STATUS PDU with  ACK_SN=7, thus acknowledging the reception of PDUs with SN=0 to SN=6,

  // and no NACK_SN provided ?

  p_RLC_Rec.AM_VTA := 7;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 32");

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU8], 0, 100);        // NOTE 2
  v_RLC_Data2 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU9], 0, 100);        // NOTE 2
  v_RLC_Data3 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU10], 0, 100);      // NOTE 2
  f_RxAMD_PDU (p_RLC_Rec,

               tsc_P_Poll,

               tsc_FI_FullSDU,

               cr_FlexPart_2LIs (100, 100),

               {v_RLC_Data1, v_RLC_Data2, v_RLC_Data3});                                    // NOTE 2
  // Step 35a: Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);
} // fl_EUTRA_7_2_3_18_TestBody


5 Execution Log Files

5.1 Samsung Tanum UE

The Samsung Tanum UE passed this test case on the Anite LTE solution with Snow 3G integrity algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file tc_7_2_3_18_Samsung-log.html:

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

6 References

	[1]
	R5s100040: This archive comprises text format execution log file.

